```
1  ----------------------------------------------------------------------------------
2  -- Name: Capt Jeff Falkinburg
3  -- Date: Spring 2016
4  -- Course: ECE 281
5  -- File: Lsn20_stoplight.vhd
6  -- HW:   Lecture 20
7  -- Purp: Basic Stoplight State Machine
8  -- Doc:  None
9  -- Academic Integrity Statement: I certify that, while others may have
10 -- assisted me in brain storming, debugging and validating this program,
11 -- the program itself is my own work. I understand that submitting code
12 -- which is the work of other individuals is a violation of the honor
13 -- code.  I also understand that if I knowingly give my original work to
14 -- another individual is also a violation of the honor code.
15 ----------------------------------------------------------------------------------
16 library IEEE;
17 use IEEE.STD_LOGIC_1164.ALL;
18
19 entity stoplight is
20     Port ( C : in  STD_LOGIC;
21            reset : in  STD_LOGIC;
22            clk : in  STD_LOGIC;
23            R : out  STD_LOGIC;
24            Y : out  STD_LOGIC;
25            G : out  STD_LOGIC);
26 end stoplight;
27
28 architecture Behavioral of stoplight is
29
30 type stoplight_state is (red, green, yellow);
31 signal next_state, current_state : stoplight_state;
32
33 begin
34
35 --next state logic
36    process (C, current_state)
37    begin
38       case current_state is
39          when red =>
40             if C = '1' then
41                next_state <= green;
42             else
43                next_state <= red;
44             end if;
45          when green =>
46             if C = '1' then
47                next_state <= green;
48             else
49                next_state <= yellow;
50             end if;
51          when yellow =>
52             next_state <= red;
53          when others =>
54             next_state <= red;
55       end case;
56    end process;
57
```

```
58    --Alternate next state logic
59    -- next_state <=  green when (current_state = red and C = '1') else
60    --                red when (current_state = red and C = '0') else
61    --                green when (current_state = green and C = '1') else
62    --                yellow when (current_state = green and C = '0') else
63    --                red when (current_state = yellow) else
64    --                red;
65
66    -- state memory
67
68       process (clk, reset) -- Only the state flip-flop needs to be in a process
69       begin                -- But we went through a case example for continuity
70          if reset = '1' then
71             current_state <= yellow;  -- Reset into yellow
72          elsif (rising_edge(clk)) then
73             current_state <= next_state;
74          end if;
75       end process;
76
77    -- output logic
78
79       R <= '1' when current_state = red else '0';
80       G <= '1' when current_state = green else '0';
81       Y <= '1' when current_state = yellow else '0';
82
83
84    end Behavioral;
85
86
```